* Syntax total physical performance score H wave.
* In this example: cut-off values based on LASA B quartiles of individual tests (in red). If you want to use cut-off values of another wave, please read the instructions in the document “How to obtain cut-off values”.

* Create walking categories.
* if not enough room/time or resp refused the test: -1.
* if resp in wheelchair or physically incapable: 0.

Compute hwalkq=$sysmis.
if (hwalk19=1 or hwalk19=2 or hwalk19=3 or hwalk19=5 or hwalk19=6) hwalkq=-1.
if (hwalk19=4 or hwalk01=2 or hwalk03=3) hwalkq=0.
if (hwalk04 ge 1 and hwalk04 le 5) hwalkq=4.
if (hwalk04 = 6 or hwalk04 = 7) hwalkq=3.
if (hwalk04 = 8 or hwalk04 = 9) hwalkq=2.
if (hwalk04 ge 10) hwalkq=1.
if (hwalk04=-2) hwalkq=-1.
Execute.

* Create chair stands categories.
* if resp cannot standup without using hands or those who did not complete the 5 rises: 0

Compute hchairq=$sysmis.
if (hchair2=1 or (hchair6 ge 0 and hchair6 le 4)) hchairq=0.
if (hchair6=5 and (hchair7 ge 1 and hchair7 le 9)) hchairq=4.
if (hchair6=5 and (hchair7 =10 or hchair7=11)) hchairq=3.
if (hchair6=5 and (hchair7=12 or hchair7=13 or hchair7=14)) hchairq=2.
if (hchair6=5 and hchair7 ge 15) hchairq=1.
if (hchair6=-2) hchairq=-1.
if (hchair6=5 and hchair7<-2) hchairq=-1.
Execute.

* Create cardigan categories.
* if resp could not complete test or only with help: 0

Compute hcardtot= sum.(hcardig4, hcardig8).
if (hcardig4<0 or hcardig8<0) hcardtot=-2.
fre hcardtot.

Compute hcardq=$sysmis.
if (hcardig1 le -1) hcardq=-1.
if (hcardig1 ge 2 or hcardig2 ge 2 or hcardig3 ge 2 or hcardig5 ge 2 or hcardig6 ge 2 or hcardig7 ge 2) hcardq=0.
if (hcardtot ge 1 and hcardtot le 8) hcardq=4.
if (hcardtot ge 9 and hcardtot le 11) hcardq=3.
if (hcardtot ge 12 and hcardtot le 15) hcardq=2.
if (hcardtot ge 16) hcardq=1.
fre hcardq.
* create tandem stand categories.
* if resp refuses test: -1.
* if resp is physically incapable: 0.

compute htandq=$sysmis.
if (htandem1=2 or htandem1=3 or htandem1=4 or htandem1=6 ) htandq=0.
if (htandem1=5) htandq=-1.
if (htandem1=1 and (htandem2 ge 3 and htandem2 le 9)) htandq=2.
if (htandem1=1 and htandem2 ge 10) htandq=4.
if (htandem1 le -1) htandq=-1.

missing value hcardq hwalkq hchairq htandq (-1).
var lab hcardq 'cardigan test: categories total time needed'.
var lab hwalkq 'walk test: categories total time needed'.
var lab hcardq 'tandem stand test: categories total time needed'.
var lab hchairq 'chair stand test: categories total time needed'.
val lab hcardq hwalkq hchairq htandq
 -1 no valid data
 0 unable to do the test
 1 slowest time category
 4 fastest time category

*Compose total physical performance score of 3 tests (lower extremity performance score, only resp with complete data):

compute hpf3q=hcardq+hwalkq+hchairq.

*Compose total physical performance score of 4 tests (only resp with complete data):

compute hpf4q=hcardq+hwalkq+hchairq+htandq.

if (sysmis(hpf3q)) hpf3q=-1.
if (sysmis(hpf4q)) hpf4q=-1.
missing value hpf3q hpf4q (-1).
var lab hpf3q 'physical performance score 3 tests'.
var lab hpf4q 'physical performance score 4 tests'.
val lab hpf3q hpf4q
 -1 invalid data on at least 1 test

fre hpf3q hpf4q.

*Higher scores on hpf3q and hpf4q indicate better physical performance.